MATHEMATICS 201-NYA-05
Differential Calculus
Martin Huard
Fall 2011

Limits and Derivatives with Maple

Limits

The evaluation of limits is done with the command

limit(***expression*** **, x =** ***a*** **)**

If you use "Limit" instead of "limit", then Maple will simply display the limit.

Example: $\lim\limits_{x\to 3}\left(x^2-x\right)$

Limit(x^2-x, x=3) = limit(x^2-x, x=3);

Where you should use some cut and paste for the second limit. Note that the first part is not absolutely needed, but is nice to have so that you can see if you entered the function correctly.

Note that you can also use Expression Palette on the left side of your screen.

One-sided limits: Use left or right.

Example: $\lim\limits_{x\to 1^-}\dfrac{1}{x^2-1}$

Limit(1/(x^2-1), x=1,left) = limit(1/(x^2-1), x=1,left);

Limits at infinity

Example: $\lim\limits_{x\to\infty}\dfrac{x}{2x-1}$

Limit(x/(2*x-1), x=infinity) = limit(x/(2*x-1), x=infinity);

Differentiation

Differentiating Expressions

The basic command to differentiate an expression is

diff(***expression*****, x)**

where the x means we are differentiating with respect to x.

For example, suppose we wish to differentiate $f(x)=x+\frac{1}{x}$.

diff(x+1/x,x);

We can simplify our answer with the **simplify** command

simplify(%);

If we need to evaluate the derivative at a point (to find, for example, the slope of the tangent line at a point) we make use of the **subs** command. For example if we want $f'(2)$ then we have

subs(x=2, diff(x+1/x,x));

For higher order derivatives $f^{(n)}(x)$ we use the "$" sign in the command: **diff(*expression*, *x*$*n*)**

For example, $f'''(x)$ is given by

diff(x+1/x, x$3);

Differentiating Functions

It is often more practical to differentiate an expression that has been introduced to Maple as a function. If we retake our example of $f(x)=x+\frac{1}{x}$ then we begin by defining our function,

f:=x->x+1/x;

The command to differentiate is now **D(f)**.

D(f);

Here the arrow notation (in the answer) indicates that the derivative is a function. To have $f'(x)$, we simply add (x) after the command.

D(f)(x);

Simplification works the same as before,

simplify(D(f)(x));

The advantage of this way is that it makes the evaluation of the function, and its derivative, at a given point a lot easier.

For example, $f(2)$ is given by

f(2);

and $f'(2)$ is given by

D(f)(2);

For higher order derivatives, we have use "@@" in the command in the following way,

(D@@*n*)(f)

For example, $f'''(x)$ is given by

(D@@3)(f)(x);

Implicit differentiation

Suppose we wish to find $\frac{dy}{dx}$ for $x^2+y^2=25$.

Defining our equation,

eq:=x^2+y^2=25;

and using the command implicitdiff(equation, y, x),

implicitdiff(eq,y,x);

Note that the order of *y* and *x* matters, since it stands for $\frac{dy}{dx}$.

To evaluate this derivative at a point, say $(3,4)$, we make use of the **subs** command.

subs(x=3, y=4, implicitdiff(eq,y,x));

For higher order derivatives, we simply add *x*'s at the end (the same number of *x* as the order of the derivative). For example $\frac{d^2y}{dx^2}$,

implicitdiff(eq,y,x,x);